
FICHA METODOLÓGICA 

NOMBRE DEL INDICADOR 
Prevalencia de desnutrición crónica en niños/as menores 

de dos años
1
 

DEFINICIÓN 

Número de niños y niñas menores de dos años de edad que 

presentan un retraso en la longitud/talla, en relación a los 

estándares esperados para su edad de acuerdo a los Patrones 

de Crecimiento Infantil establecidos por la Organización Mundial 

de la Salud (OMS); expresado como porcentaje de niños y niñas 

menores de 2 años. 

FÓRMULA DE CÁLCULO 

 

𝑃𝐷𝐶𝑁<2𝑎ñ𝑜𝑠 =
𝑁𝐷𝐶<2𝑎ñ𝑜𝑠

𝑇𝑁<2𝑎ñ𝑜𝑠  
∗ 𝐾 

Donde: 

 

𝑃𝐷𝐶𝑁<2𝑎ñ𝑜𝑠= Prevalencia de la desnutrición crónica en niños/as menores de dos años en el periodo 

determinado (t). 

 

𝑁𝐷𝐶<2𝑎ñ𝑜𝑠 = Número de niños menores de dos años de edad que presentan desnutrición crónica en un año 

o periodo determinado (t). 

 

𝑇𝑁<2𝑎ñ𝑜𝑠 = Total de niños y niñas menores de dos años en el periodo determinado (t). 

 

K = 100 

 

DEFINICIÓN DE LAS VARIABLES RELACIONADAS 

 

Desnutrición.- Estado patológico resultante de una dieta deficiente en uno o varios nutrientes esenciales 

o de una mala asimilación de los alimentos. (Glosario de Términos de Desnutrición-UNICEF, 2012) 

 

Desnutrición Crónica.- La desnutrición crónica se ve reflejada en el crecimiento alcanzado en longitud o 

talla para la edad del niño en una visita determinada. El retardo en el crecimiento de los niños conocido 

como desnutrición crónica (longitud o talla, baja) se debe a un prolongado aporte insuficiente de nutrientes 

o enfermedades infecciosas recurrentes. (Patrones de Crecimiento para el niño-OMS, 2008; pág.3)    

 

Longitud o Talla.- Es la medida que refleja la estatura o altura de las personas. Para todos los 

indicadores que incluyen longitud o talla, longitud acostada debe ser usada para niños menores de 24 

meses y talla parada, para niños de 24 meses y mayores. El indicador de desnutrición crónica se mide en 

centímetros (cm). (Manual del ANTHRO -OMS, 2007) 

 

METODOLOGÍA DE CÁLCULO 

 

Este indicador se obtiene de dividir el número de personas menores a 2 años de edad que tienen 

                                                 
1
 La Comisión Especial de Estadísticas de Salud, mediante reunión mantenida, el día jueves 07 de mayo de 2015, 

resolvió a partir del acuerdo CEES 006-2015, aprobar la metodología de cálculo del indicador de la tasa de desnutrición 
en sus diferentes tipos: crónica, aguda y global. 



desnutrición crónica en un año o periodo determinado (t) entre la población del mismo grupo de edad, en 

un año o periodo determinado (t), multiplicado por cien. 

 

En el caso de la Encuesta Nacional de Salud y Nutrición (2018), se selecciona la sección “antropometría”, 

y se obtiene la desnutrición crónica de la siguiente manera: 

 

 Para la toma del peso, longitud y talla se usan dos mediciones y en el caso de haber una 

diferencia de ± 0.5 kg para peso y ± 0.5 cm para talla, se toma una tercera medición. El valor final 

es la media entre las dos mediciones. En caso de haber una tercera medición, esta representa la 

media de las dos mediciones más cercanas. 

 

 Una vez que se tienen calculadas las variables edad en días, peso, longitud y talla se procede a 

estandarizar los valores longitud o talla para la edad, peso para la longitud o talla y peso para la 

edad en base al Manual Anthro (OMS-2007; pág. 20-24) y a las tablas de patrones de crecimiento 

de la OMS. (OMS-2006; Child Growth Standars). 

 

 Para el cálculo de la desnutrición crónica se toma, exclusivamente, la variable estandarizada de 
longitud o talla para la edad. En este sentido, las personas menores de dos años de edad con 
retardo en longitud/talla son registradas como las que tienen un puntaje estandarizado menor a -
2.0 desviaciones estándar. En este procedimiento se corrigen los valores extremos para este 
indicador; así se eliminan los valores que están fuera del intervalo -6.0 y +6.0 desviaciones 
estándar. (Tabla 1.) 

 
 

Tabla 1. Valores plausibles de los indicadores 

antropométricos en niños menores de 2 años de edad 

 

Indicador Criterios o variables 
antropométricas 

Puntajes Z 

Desnutrición Crónica 
Longitud o talla para la 
edad  

-6.0 y +6.0 

Desnutrición Global Peso para la edad -6.0 y +5.0 

Desnutrición Aguda 
Peso para la 
talla/longitud 

-5.0 y +5.0 

 
Fuente: OMS, 2006 

 

 Se toma en cuenta la variable estandarizada de longitud según la edad, para el cálculo de la 

desnutrición crónica. En este sentido las personas menores de 2 años de edad con retardo en talla 

son registradas como las que tienen un puntaje estandarizado menor a -2.0 desviaciones 

estándar. En este procedimiento se corrigen los valores extremos para este indicador 

 

 Para el cálculo del indicador de desnutrición crónica se toma en cuenta los puntajes 

estandarizados para las personas menores de 2 años, edad que corresponde a 24 meses. Según 

el Manual Anthro de la OMS, 24 meses equivalen a 730,5 días. 

 

LIMITACIONES TÉCNICAS 

En encuestas previas a la Encuesta Nacional de Salud y Nutrición (ENSANUT, 2012), únicamente se 

registró una sola medición para la toma de talla y peso, por lo tanto no se puede aplicar la metodología de 

las tres mediciones que emplean en la ENSANUT o en la Encuesta de Condiciones de Vida - ECV 2014. 

 

UNIDAD DE MEDIDA O EXPRESIÓN DEL 

INDICADOR 

 

Porcentaje 

 



INTERPRETACIÓN DEL INDICADOR 

En el período (t), por cada 100 niños/as menores 

de 2 años de edad se estiman XX niños con 

desnutrición crónica. 

FUENTE DE DATOS 

Instituto Nacional de Estadística y Censos (INEC) 

y Ministerio de Salud Pública: Encuesta Nacional 

de Salud y Nutrición, 2012 (ENSANUT). 

 

Instituto Nacional de Estadística y Censos 

(INEC): Encuesta de Condiciones de Vida, 2014 

(ECV). 

 

Instituto Nacional de Estadística y Censos 

(INEC): Encuesta Nacional de Salud y Nutrición, 

2018 (ENSANUT). 

PERIODICIDAD DEL INDICADOR  Indeterminada 

DISPONIBILIDAD DE LOS DATOS  2012, 2014 y 2018 

CONSEJO SECTORIAL Y/O INSTITUCIÓN 

RESPONSABLE DE LA TRANSFERENCIA DE LA 

INFORMACIÓN 

Consejo Sectorial de lo Social -  Ministerio de 

Salud Pública (MSP).

FECHAS DE TRANSFERENCIA DE LA 

INFORMACIÓN 

En función de la publicación de la Encuesta de 

Salud y Nutrición y otras fuentes de información 

que recolecten variables antropométricas. 

NIVEL DE 

DESAGREGACIÓN 

GEOGRÁFICO Nacional, urbano y rural
3
 

GENERAL No aplica 

OTROS ÁMBITOS No aplica 

INFORMACIÓN GEO – REFERENCIADA No aplica 

RELACIÓN CON INSTRUMENTOS DE 

PLANIFICACIÓN NACIONAL E INTERNACIONAL 

Plan Nacional de Desarrollo 2017-2021 

Objetivo 1: Garantizar una vida digna con 

iguales oportunidades para todas las personas. 

 

Meta: Reducir de 24,8% al 14,8%  la prevalencia 

de desnutrición crónica en niños menores a 2 

años, al 2021 

 

Agenda de Desarrollo 2030 - Objetivos de 

Desarrollo Sostenible (ODS). 

 

Objetivo 2: Poner fin al hambre, lograr la 

En atención a los procesos de reestructura de la Administración Pública, aquellas entidades que asuman las 
competencias de las registradas como responsables, asumirán las responsabilidades en cuanto al reporte de este 
indicador. 
3
 Los demás niveles de desagregación, estarán en función al procesamiento de la información, y a los resultados 

obtenidos en la aplicación de los parámetros establecidos por el INEC, de acuerdo a los documentos técnicos emitidos 
según la operación estadística pertinente. 



seguridad alimentaria y la mejora de la nutrición y 

promover la agricultura sostenible. 

 

Meta 2.2: Para 2030, poner fin a todas las formas 

de malnutrición, incluso logrando, a más tardar 

en 2025, las metas convenidas 

internacionalmente sobre el retraso del 

crecimiento y la emaciación de los niños menores 

de 5 años, y abordar las necesidades de nutrición 

de las adolescentes, las mujeres embarazadas y 

lactantes y las personas de edad 

 

Indicador 2.2.1: Prevalencia del retraso en el 

crecimiento (estatura para la edad, desviación 

típica < -2 de la mediana de los patrones de 

crecimiento infantil de la Organización Mundial de 

la Salud (OMS)) entre los niños menores de 5 

años 

 

REFERENCIAS BIBLIOGRÁFICAS DE LA 

CONSTRUCCIÓN DEL INDICADOR 

Fondo de las Naciones Unidas para la Infancia 

(UNICEF) “Glosario de Términos de 

Desnutrición”. 

 

Organización Mundial de la Salud (OMS 2008) 

“Patrones de Crecimiento para el niño”. 

 

Organización Mundial de la Salud (OMS 2007) 

“Manual WHO Anthro para computadoras 

personales” Software para evaluar el crecimiento 

y desarrollo de los niños del mundo.  

 

Organización Mundial de la Salud (OMS 2006) 

“Child Growth Standars” 

 

Instituto Nacional de Estadística y Censos (2012): 

“Encuesta Nacional de Salud y Nutrición” 

ENSANUT, 2012. Tomo I. 

FECHA DE ELABORACIÓN DE LA FICHA 

METODOLÓGICA 
Abril, 2013 

FECHA DE LA ÚLTIMA ACTUALIZACIÓN DE LA 

FICHA 
Diciembre de 2019 

CLASIFICADOR SECTORIAL Salud 1.4  

HOMOLOGACIÓN DEL INDICADOR Homologado 

FECHA DE HOMOLOGACIÓN DEL INDICADOR 14 de agosto de 2017 

RESEÑA DEL INDICADOR No aplica 

ELABORADO POR 
Dirección de Estadísticas  Sociodemográficas- 

Instituto Nacional de Estadística y Censos. 



 

Responsables: 

Analista: José Andrade 

Director: Andrés Albán 

 

Aprobado: Mesa Temática de Salud 

 

ANEXOS 

Anexo 1. Algoritmo del Cálculo del Indicador 

SINTAXIS STATA 

*==============================================================================* 

* TÍTULO DE LA SINTAXIS:          

       

* Prevalencia de desnutrición crónica en niños/as menores de dos años    

            

* OPERACIÓN ESTADÍSTICA:          

       

* Encuesta Nacional de Salud y Nutrición (ENSANUT - 2018)     

       

* UNIDAD TÉCNICA RESPONSABLE:          

       

* Instituto Nacional de Estadística y Censos (INEC) 

        

*==============================================================================* 

* Fecha de elaboración: Diciembre del 2018       

     

* Fecha última modificación: Diciembre del 2018      

                 

*==============================================================================* 

*==============================================================================* 

* Instrucciones: 

* 1. Descargar las bases que tengan las variables antropometricas en formato .dta 

* 2. Guardar en un directorio y copiar ese directorio en "cd" 

* 3. Correr la programación en stata 

* 4. Es necesario que exista conexión a internet para descargar el paquete  

*  Child growth standards "igrowup_stata". 

*==============================================================================* 

clear all 

set more off 

* En esta sección se copia la dirección en donde se encuentran las bases de datos 

* que contienen las variables de antropometría (Ej: cd "C:\...") 

cd "C:\ENSANUT 2018\Bases de datos\Bases_trabajadas" 

global dir2 = "C:\Desnutricion\igrowup" 

global dir3 = "C:\sobrepeso\Who2007" 

global dirr = "C:\ENSANUT 2018\Resultados" 

 

use 1_BDD_ENS2018_f1_personas, clear  

 

merge m:1 id_hogar using 2_BDD_ENS2018_f1_hogar 

keep if _merge==3 

drop _merge 

 

svyset upm [iw=fexp], strata(estrato) vce(linearized) singleunit(certainty)  

 

*==============================================================================* 

* Cálculo de variables antropométricas necesarias para el indicador 

*==============================================================================* 

*** 1.- Estimación de la edad (días) 

cap drop edaddias  

** Fecha de nacimiento 

gen dob=mdy(f1_s7_2_2,f1_s7_2_1,f1_s7_2_3) 



** Fecha de medición  

gen dov=mdy(f1_s7_3_2,f1_s7_3_1,f1_s7_3_3) 

** Edad en días 

gen edaddias= dov-dob 

 

*** 2.- Estimación del peso (kg) 

* Validación de las 3 medidas del peso 

gen aux = f1_s7_4_1 - f1_s7_4_2 

replace aux = aux*-1 if aux<0 

replace f1_s7_4_3=. if aux<=0.5 & f1_s7_4_3!=. 

drop aux 

 

* Se calcula el peso en kg. 

gen peso=. 

* Distancia entre las tres medidas 

gen d1 = abs(f1_s7_4_1 - f1_s7_4_2)  

gen d2 = abs(f1_s7_4_1 - f1_s7_4_3)  

gen d3 = abs(f1_s7_4_2 - f1_s7_4_3)  

* Variable identificador si distancia entre toma 1 y toma 2 menor o igual a 0.5 

gen s = 0 if f1_s7_4_1!=. 

replace s = 1 if d1  <= 0.5 

* Promedio simple entre toma 1 y toma 2 

replace peso =(f1_s7_4_1+f1_s7_4_2)/2 if s==1 

* Caso contrario, promedio de la menor distancia entre las 3 mediciones 

* distancia mínima 

gen dmin= min(d1,d2,d3) 

replace peso = (f1_s7_4_1 + f1_s7_4_2)/2 if d1==dmin 

replace peso = (f1_s7_4_1 + f1_s7_4_3)/2 if d2==dmin 

replace peso = (f1_s7_4_2 + f1_s7_4_3)/2 if d3==dmin 

drop d1 d2 d3 s dmin 

 

*** 3.- Estimación de la talla (cm) 

* Validación de las 3 medidas de la longitud 

gen aux = f1_s7_5_1 - f1_s7_5_2 

replace aux = aux*-1 if aux<0 

replace f1_s7_5_3=. if aux<=0.5 & f1_s7_5_3!=. 

drop aux 

* Validación de las 3 medidas de la talla 

gen aux = f1_s7_6_1 - f1_s7_6_2 

replace aux = aux*-1 if aux<0 

replace f1_s7_6_3=. if aux<=0.5 & f1_s7_6_3!=. 

drop aux 

 

* Se calcula la talla en cm. 

gen talla =. 

* Consolido las tomas para longitug y talla 

egen talla1 = rowtotal(f1_s7_5_1 f1_s7_6_1), missing 

egen talla2 = rowtotal(f1_s7_5_2 f1_s7_6_2), missing 

egen talla3 = rowtotal(f1_s7_5_3 f1_s7_6_3), missing 

* Distancia entre las tres medidas 

gen d1 = abs(talla1- talla2) 

gen d2 = abs(talla1- talla3) 

gen d3 = abs(talla2- talla3) 

* Variable identificador si distancia entre toma 1 y toma 2 menor o igual a 0.5 

gen s = 0 if talla1 !=. 

replace s = 1 if d1  <= 0.5 

* Promedio simple entre toma 1 y toma 2 

replace talla =(talla1+talla2)/2 if s==1 

* Caso contrario, promedio de la menor distancia entre las 3 mediciones 

* distancia mínima 

gen dmin= min(d1,d2,d3) 

replace talla = (talla1+talla2)/2 if d1==dmin 

replace talla = (talla1+talla3)/2 if d2==dmin 

replace talla = (talla2+talla3)/2 if d3==dmin 

drop d1 d2 d3 s dmin 

 

*** 4. Medida de talla o longitud 

* H: Talla - L: Longitud  



gen medida="H" if f1_s7_6_1!=. 

replace medida="L" if f1_s7_5_1!=. 

 

*==============================================================================* 

* Utilización del código de desnutrición de ANTHRO - OMS (igrowup) 

*==============================================================================* 

*Crear un nuevo directorio "igrowup": 

local y ="$dir2" 

cap mkdir "`y'" 

cd "`y'" 

adopath + "`y'" 

*==============================================================================* 

* Nota: En caso de no contar con el paquete, a continuación se encuestran los códigos 

* de descarga: 

*Descarga y extracción del paquete de Child growth standards "igrowup_stata": 

*cap copy http://www.who.int/childgrowth/software/igrowup_stata.zip /// 

*  igrowup_stata.zip 

*cap unzipfile igrowup_stata.zip 

*cd "`dir'" 

*==============================================================================* 

*Generar los tres primeros parámetros: reflib, datalib y datalab. 

gen str200 reflib="`y'" 

lab var reflib "Directory of reference tables" 

gen str200 datalib="`y'" 

lab var datalib "Directory for datafiles" 

gen str30 datalab="f1_sn" 

lab var datalab "Working file" 

 

* Se generan las variables para la macro "igrowup". 

gen sex=sexo 

gen agemons=edaddias 

gen str6 ageunit="days" 

lab var ageunit "days" 

gen weight=peso 

gen height=talla 

gen str1 measure=" " 

gen headc=. 

gen muac=. 

gen triskin=. 

gen subskin=. 

gen sub=. 

gen str1 oedema="n" 

gen sw=-1 

 

* Completar en la macro los parámetros para que corra el comando 

igrowup_standard reflib datalib datalab sex agemons /// 

ageunit weight height measure head muac tri sub oedema sw 

drop reflib datalib datalab 

*==============================================================================* 

* Construcción de las variables de desnutrición con los zscore 

** Definición de la edad en días de la población menor a 2 años: 

** Días= 365.25 * 2 = 730.5 ----> 731 (valor aproximado) (Manual Anthro OMS) 

*==============================================================================* 

* T7_i1_2. Desnutrición crónica para menores de 2 años de edad 

*==============================================================================* 

cap drop dcronica_2  

gen  dcronica_2=0 if _zlen>=-6 & _zlen<=6 // no se consideran outliers. 

replace dcronica_2=100 if (_zlen!=. & agemons<731 & _zlen< -2) & _zlen>=-6 & _zlen<=6 

replace dcronica_2=. if (agemons>=731) 

replace dcronica_2=. if _zlen==. 

rename dcronica_2 T7_i1_2 

 

**** Resultados *** 

mean T7_i1_2 [iw=fexp] if agemons<731 & _zlen!=. 

 

*====================================== FIN =====================================* 

 

 


